
Westsächsische Hochschule Zwickau (FH)

Fachbereich Physikalische Technik/Informatik

                  Prof. Dr. Rolf  U r b a n

               Dipl. Inf. Heiko Schellhorn

Tk-Grundlagen nach Grote

  grafisches Toolkit 

  Tcl-Befehle, um Widgets zu erzeugen und zu manipulieren 

  Geometrie-Manager 

  ereignisgesteuert ( event driven ) 

  Interaktion mit Window-Manager 

Beispiel

button .hello -text "Hello" -command { puts "Hello world" }

pack .hello

Widget-Namen

Widgets sind hierarchisch organisiert. 

Als Trennzeichen dient der Punkt. 

Ihre Namen dürfen nur aus Kleinbuchstaben und Ziffern bestehen. 

Beispiel

frame .f

button .f.b1

Widgets erzeugen und konfigurieren

Ein Widget wird mit 

widgetclass  name  ?option  value?  .....   

erzeugt. Als Ergebnis des Befehls wird der Widgetname zurückgegeben. 

Optionen sind abhängig von der Widgetklasse. Wenn sie nicht angegeben werden, 

wird ein Defaultwert verwendet. 

Wenn ein Widget schon existiert, können seine Optionen mit 

widgetname  configure  ?option value? .... 

geändert werden. 

configure kann als config abgekürzt werden. 

Ohne Optionen wird die aktuelle Konfiguration ausgegeben. 

Einzelne Optionen können mit 

widgetname  cget  option 

erfragt werden. 

Beispiel

set B [button .hello -text "Hello" -command { puts "Hello world" }]

pack $B

.hello config -foreground blue -text "Hi!"

puts [$B cget -command]

Buttons

Buttons (Schalter) gehören zu den am häufigsten verwendeten Widgets. 

Wenn der Button gedrückt wird, wird der zugehörige Befehl ausgeführt. 

button  name  ?option value? 

-command  befehl 

verbindet befehl  mit dem Button. 

befehl  wird im globalen Namensraum ausgeführt ! 

Beschriftung

Der Text, der auf einem Button steht, kann auf zwei Arten angegeben werden: 

-text   text      

gibt den Text direkt an. 

-textvariable  var     
 verwendet den Inhalt der Variablen  var . 

Beispiel

button .b -text "Drück mich!" -textvariable txt    -command { set txt "Nochmal bitte" }  

Textgestaltung

Eine Reihe von Optionen erlauben es, die Art der Beschriftung 
eines Buttons zu verändern: 
  

-font  font 



Die Schriftart font  verwenden 

-width  breite 



Breite in Zeichen 

-height  höhe 



Höhe in Zeichen 

-anchor  center|n|ne|e|se|s|sw|w|nw 

Ausrichtung des Textes im Button 

-wraplength  n 



Umbruch nach n Bildschirmeinheiten . 0 bedeutet kein Umbruch 

-justify   center|right|left 
Ausrichtung des Textes bei Umbruch 

-padx breite 



Zusätzliche Pixel in horizontaler Richtung 

-pady höhe 



Zusätzliche Pixel in vertikaler Richtung 
 -underline  index 


Zeichen an der Stelle  index wird unterstrichen 


Bildschirmeinheiten (Pixel)  können auch in 


      c    Zentimeter 

      m    Millimeter 

      i    Inch 

      p    Printer Points ( 1/72 inch ) 
 

angegeben werden. 
  


  

Beispiel

button .b -text Zentral -width 20 -height 5

button .c -text Südost -width 20 -height 5 -anchor se

set lang "Dies ist ein ziemlich langer und sinnloser Text"

button .d -textvariable lang -underline 0

button .e -textvariable lang -wraplength 2c

button .f -textvariable lang -wraplength 2c -justify right

button .g -text Quit -font { times 14 { bold italic }} -command exit

button .h -text Quit -font { courier 10 { bold italic }} -padx 10m -pady 5m

pack .b .c .d .e .f .g .h

Stil

Das Aussehen eines Button kann mit den folgenden Optionen 
verändert werden:  

-borderwidth   width 

gibt die Breite der Kanten des Buttons an. 

-borderwidth kann mit -bd abgekürzt werden. 

-relief    raised|flat|groove|ridge|sunken 

gibt an, wie die Kanten dargestellt werden. 

Wenn der Button gedrückt wird, ist das Relief in jedem  all sunken. 
  

Beispiel

foreach relief { flat raised sunken groove ridge } {

    button .$relief -text $relief -relief $relief

    button .4$relief -text $relief -relief $relief -borderwidth 4

    button .8$relief -text $relief -relief $relief -borderwidth 8

    grid .$relief .4$relief .8$relief -padx 5 -pady 5

} 

Farben

können auf zwei Arten angegeben werden: 

- als symbolischer Name (red, lightblue4, grey75) 

- als hexadezimaler Farbwert:  #RRGGBB 

Mit tk_chooseColor kann man Farbwerte am Bildschirm ermitteln.  

Farbattribute

-background (-bg)  farbe 
intergrundfarbe 

-foreground (-fg)  farbe 

Vordergrundfarbe 

-activebackground   farbe 
intergrundfarbe, wenn der Mauszeiger über dem Button ist 

-activeforeground   farbe 
Vordergrundfarbe, wenn der Mauszeiger über dem Button ist 

Beispiel

button .b -text Button \

        -activebackground    yellow \

        -activeforeground    red \

        -background          cyan \

        -foreground          Black  \

        -command {

                set farbe [tk_chooseColor]

                if {$farbe == ""} exit

                .b config -bg $farbe -text $farbe

        }

Bilder auf Buttons

Auf einem Button können außer Texten auch Grafiken stehen: 
 

-bitmap bitmap 

beschriftet den Button mit bitmap. 

bitmap ist entweder eine Builtin-Bitmap oder eine Datei 

in der Form @bitmapfile. 

  

-image image 

stellt das Bild image auf dem Button dar. 
  

Beispiel

# Builtin-Bitmaps

foreach name {error gray12 gray50 hourglass \

              info questhead question warning} {

    button .$name -bitmap $name 

    pack .$name -side left -padx 5

}

Cursor

-cursor  cursor 

erlaubt es, das Aussehen der Mauszeigers zu verändern, 

wenn dieser über dem Button ist. 

Beispiel

proc newCursor Widget {

    global Cursors Index Anzahl

    incr Index

    if {$Index == $Anzahl } { set Index 0 }

    set cur [lindex $Cursors $Index]

    $Widget configure -cursor $cur -text $cur

}

set Cursors {

    X_cursor target plus xterm circle man gumby mouse

    spider spraycan pencil arrow center_ptr

}

set Index 0

set Anzahl [llength $Cursors]

button .c -text [lindex $Cursors 0] -command {newCursor .c} \

          -width 25 -height 5 -anchor n

Tastatureingaben
Buttons können auch über die Tastatur bedient werden. 
Dabei  wechselt man mit  <TAB> und  <Shift-TAB>  zwischen 
den einzelnen Buttons. Mit  <SPACE> wird der Button aktiviert. 

-highlightcolor  color 

Farbe der Rahmens, wenn der Button Tastatureingaben annimmt 

 -highlightbackground  color 
Farbe der Rahmens, wenn der Button  keine Tastatureingaben annimmt  

-highlightthickness  pixel 
Dicke der Rahmens   

-takefocus   1|0 

Tastatureingaben erlauben (1) bzw. verbieten (0) 
  

Beispiel

 

button .a -text  "def def 1" -command { puts A }

button .b -text  "red blue 1" -command { puts B } \

    -highlightcolor red -highlightbackground blue

button .c -text  "red blue 3" -command { puts C } \

    -highlightcolor red -highlightbackground blue \

    -highlightthickness 3 -takefocus 0

button .d -text  "red def 3" -command { puts D } \

    -highlightcolor red -highlightthickness 3

Buttons deaktivieren
Manchmal ist es nötig, bestimmte Funktionen einer grafischen Oberfläche 
zeitweise zu deaktivieren.   

-state   normal|disabled 
aktiviert bzw deaktiviert den Button   

-disabledforeground  color 
Vordergrundfarbe, wenn der Button inaktiv ist 
  

Beispiel

button .b -text Display -command { puts Irgendwas } \

        -disabledforeground yellow 

button .d -text Disable -command { .b config -state disabled }

button .e -text Enable -command { .b config -state normal }

  

Buttons vom Programm aus bedienen
Es ist auch möglich, einen Button vom Programm aus 
aufzurufen: 
$button  invoke 
ruft den zu  $button gehörigen Befehl auf 

$button  flash 

lässt  $button  blinken 
  

Beispiel

button .b -text Flash -command { .c flash }

button .c -text Call  -command { .b invoke }

Geometrie-Manager

Ein Widget wird in Tk nur dann auf dem Bildschirm dargestellt, wenn es 
von einem Geometrie-Manager verwaltet wird. 

Geometrie-Manager ordnen Widgets in einem übergeordneten Fenster 
( parent widget ) an. Dieses ist normalerweise das Basisfenster ( root 
window ) oder ein frame-Widget. 

Zu Tk gehören drei verschiedene Geometrie-Manager:  

      pack 

      grid 

      place 

place  arbeitet mit absoluten oder relativen Bildschirmkoordinaten und 
wird nur für spezielle Aufgaben verwendet. 

pack

pack ist der ursprüngliche Geometrie-Manager von Tk. 

Um Widgets auf dem Bildschirm zu positionieren, verwendet pack 
keine absoluten Koordinaten, sondern ein Rezept, das angibt, 
wie die Widgets angeordnet werden sollen.  

Beispiel

pack .b1 .b2 

pack .b1 .b2 -side bottom 

pack .b1 .b2 -side left 
  

Der Pack-Algorithmus
pack arbeitet nach folgendem Verfahren: 

1. Widgets werden im freien Bereich ( cavity ) ihres parents angelegt. 
    Jedes Widget belegt dabei die ganze Seite des freien Bereichs, 
    an der es gepackt wird. 

2. In dieses Rechteck (parcel, allocation rectangle) wird das Widget 
    plaziert. 

3. Für weitere Widgets steht nur noch der Platz außerhalb des Rechtecks 
    zur Verfügung. 

4. Nachdem alle Widgets angeordnet sind, werden sie auf die natürliche 
    Größe konfiguriert (shrink wrapping). 

 

Beispiel

freier Bereich weiss,    Rechteck grau,     Widgets bunt 

        Nach Anlage eines Buttons 

       3 Buttons ( top, left, bottom ) 

            nach shrink-wrapping 
  

Reihenfolge

Bei dem oben beschriebenen Verfahren ist die Reihenfolge der pack-Befehle entscheidend. 
  
Diese bestimmt außerdem, wie Widgets beim Verkleinern des Fensters verschwinden: 
Das zuletzt gepackte Widget wird zuerst wieder entfernt. 

Für die Funktion einer grafischen Oberfläche wichtige Widgets 
sollten daher zuerst gepackt werden.   

Beispiel

     3 Buttons ( top, bottom, left ) 

                               nach shrink-wrapping 
  
Filling

Wenn ein Widget seine natürliche Größe hat, kann es sein, dass 
es nicht das ganze zugehörige Rechteck belegt. Mit der Option 

-fill  style 
legt man fest, was in diesem Fall geschehen soll: 

none 

Das Rechteck wird nicht belegt (Default). 

x 

Das Widget dehnt sich horizontal so weit wie möglich aus. 

y 

Das Widget dehnt sich vertikal so weit wie möglich aus. 

both 

Das Widget dehnt sich so weit wie möglich aus. 
  

Beispiel

    pack .b1

    pack .b1 -fill x


Fenster vergrößern

Beim Vergrößern eines Fensters geschieht normalerweise nichts 
mit den im Fenster befindlichen Widgets. Jedes Widget bleibt an 
der Kante, an der es gepackt wurde. 
  
Mit der Option 
-expand true 
erreicht man, dass sich das zu einem Widget gehörige Rechteck in 
sonst unbelegtem Platz im Fenster ausdehnt. 

-expand false bewirkt das oben beschriebene Default-Verhalten. 

(Statt true und false kann man auch 1 und 0 angeben.) 

In der Regel wird man zusammen mit -expand true auch -fill both angeben.  

Beispiel

button .top -text Topleiste -bg white

button .exit -text Exit -bg green -command {exit}

button .exp -text true -bg yellow

button .flaeche -text Fläche -bg lightblue

button .egal -text Egal -bg lightgreen

pack .top -fill x

pack .flaeche -expand true -fill both -side bottom

pack .egal .exp -side left

pack .exit -side right

        natürliche Größe 

        nach Vergrößerung 
  

Anchoring und Padding

Mit den folgenden Optionen kann das Verhalten und Aussehen 
eines Widgets geändert werden: 

-anchor  center|n|ne|e|se|s|sw|w|nw 

gibt an, wo im zugehörigen Rechteck das Widget positioniert wird. 

  
-ipadx breite 
-ipady höhe 
fügen zusätzlichen Platz im Widget hinzu. 

  
-padx breite 
-pady höhe 
fügen zusatzlichen Platz zwischen Widget und Rechteck ein. 
  

Beispiel

button 
.b1 -text bpadx -anchor e -padx 20 -bg white 

pack 
.b1  -pady 5

button 
.b2 -text ipadx -anchor e -bg white 

pack 
.b2 -ipadx 20 -pady 5

button 
.b3 -text padx -bg white -width 20

pack 
.b3 -padx 30  -pady 5

button 
.b4 -text Exit -bg white -command exit

pack 
.b4  -pady 5 -anchor w

 
[image: image20.png]



  

Informationen vom Packer
pack slaves master 


liefert eine Liste der in master gepackten Widgets, 

pack info widget 


eine Liste mit den Pack-Optionen von widget.  

pack propagate master ?boolean?  verhindert, dass der Shrink-Wrap-Schritt des Packers 

für master ausgeführt wird, wenn boolean  false ist. 


Beispiel

button .b -text B -command exit

pack .b

puts [ pack info .b ]

puts [ pack slaves . ]

Der update-Befehl

Tcl/Tk verarbeitet Events normalerweise nur dann, wenn nichts 
anderes zu tun ist. Insbesondere werden Änderungen an Widgets 
und Benutzereingaben nicht unmittelbar bearbeitet, was bei länger 
laufenden Programmen den Eindruck erweckt, die Anwendung sei 
abgestürzt. 

Abhilfe schafft der Befehl 

update    . 

Er verarbeitet sämtliche Befehle aus der Event-Queue und danach sämtliche Bildschirm-Updates. 
Wenn man nur ein Auffrischen des Bildschirms erreichen will, sollte man 

update idletasks 

verwenden. 


Beispiel

label .count -text Ready

button .fire -text Fire -command fire

pack .count .fire -padx 4 -pady 4

proc fire {} {

    for { set i 5 } { $i >= 0 } { incr i -1 } {

        .count config -text "$i sec"

        update idletasks

        after 1000

    }

    .fire flash

    .count config -text Ready

}

Widgets umpacken und entpacken

Die Optionen 

-before otherwidget    und 
-after otherwidgeht 

bewirken, dass ein Widget vor bzw. hinter einem anderen eingefügt wird. 

Der Befehl 

pack forget  widget ?widget2?  ... 

entfernt gepackte Widgets wieder. Auch wenn ein Widget nicht gepackt 
ist, kann es trotzdem umkonfiguriert werden. 
   

Beispiel

button .a -text A -command {.q config -fg red }

button .b -text B

button .q -text Quit -command exit 

button .d -text Disable -textvariable EnDis \

          -command { endisable .b .q }

pack .a .b .q .d -side left

update idletasks

pack propagate . false 

proc endisable { args }

    global EnDis

    if { $EnDis == "Disable" } {

         set cmd "pack forget $args"

         set EnDis "Enable"

    } else {

         set cmd "pack $args -side left -before .d"

         set EnDis "Disable"

    }

    eval $cmd

}


Frames

Nur mit den Mitteln von pack lässt sich das folgende Fenster 
nicht erzeugen:   
  
[image: image21.png]5 T 7]

Schalter_A | Schalter_8

|_Schater.c




Zur Lösung des Problems verwendet man ein frame. 
Frames (Rahmen) sind Widgets, die hauptsächlich zur Aufnahme 
anderer Widgets dienen. 
  

Beispiel

frame .top 

button .top.a -text Schalter_A -background white

button .top.b -text Schalter_B -background lightblue

pack .top.a .top.b -side left

pack .top -side top

frame .bot

button .bot.c -text Schalter_C -background lightgreen

button .bot.d -text Schalter_D -background yellow

pack .bot.c .bot.d -side left

pack .bot -side top

Das Frame-Widget

Entsprechend seiner Funktion hat das Frame-Widget nur wenige Optionen: 

-background (-bg)  farbe 
-borderwidth   width 
-cursor  cursor 
-height  höhe 
-highlightbackground  color 
-highlightcolor  color 
-highlightthickness  pixel 
-relief   flat|raised|sunken|groove|ridge 
-takefocus   1|0 
-width  breite 

Außer zur Zusammenfassung von Widgets können Frames auch gut zur Erzeugung von Trennlinien verwendet werden. 
Beispiel

label .l -text "Verwendung von Frames" 

pack .l -side top -fill x -pady 10

frame .sep -height 3 -borderwidth 1 -relief sunken

pack .sep -side top -fill x -pady 5

frame .buttons -background lemonchiffon 

button .buttons.a -text A 

button .buttons.b -text B

button .buttons.q -text Quit -command exit 

pack .buttons.a .buttons.b .buttons.q -side left -padx 4

button .d -text Disable -textvariable EnDis \

          -command { endisable .buttons }

pack .buttons -side left -ipady 6 -ipadx 6

pack .d -side right -padx 5

update idletasks

pack propagate . false

proc endisable { args } {

    global EnDis

    if { $EnDis == "Disable" } {

        set cmd "pack forget $args"

        set EnDis "Enable"

    } else {

        set cmd "pack $args -side left -before .d \

                            -ipady 6 -ipadx 6"

        set EnDis "Disable"

    }

    eval $cmd

}

Label

Zur Beschriftung einer grafischen Oberfläche dient das label-Widget. 
Es erzeugt einen nicht-interaktiven Text. 

label  labelname ?optionen? . 

Der Text eines Labels kann auf zwei Arten angegeben werden:  

-text  text 


gibt den Text direkt an.   

-textvariable  var 

verwendet den Inhalt der Variablen  var . 


Wenn der Text statisch ist, sollte man -text verwenden. 
Wenn er sich häufig ändert, ist -textvariable einfacher. 
  

Beispiel

label .l_status -text "Status: " 

label .status -textvariable status \

        -relief sunken -borderwidth 4 -pady 3 -padx 3

pack .l_status .status -side left

set status "Initializing ...."

button .b -text Go -command { set status "Done." }

button .q -text quit -command exit

pack .q .b -side bottom -padx 10 -pady 5

Optionen für Label-Widgets

Außer -text und -textvariable gibt es noch die folgenden Konfigurationsmöglichkeiten für Label: 

-anchor  center|n|ne|e|se|s|sw|w|nw 
-background (-bg)  farbe 
-bitmap bitmap 
-borderwidth   width 
-cursor  cursor 
-disabledforeground  color 
-font  font 
-foreground (-fg)  farbe 
-height  höhe 
-highlightbackground  color 
-highlightcolor  color 
-highlightthickness  pixel 
-image image 
-justify   center|right|left 
-padx breite 
-pady höhe 
-relief flat|raised|sunken|groove|ridge 
-state   normal|disabled 
-takefocus   0|1 
-underline  index 
-width  breite 
-wraplength  pixel 

Sie wirken genauso wie bei Buttons. 
Zu beachten ist, dass -wraplength in Bildschirmeinheiten angegeben wird, 

-width dagegen für Texte in Zeichen. 

Messages

Eine andere Möglichkeit, einen Text darzustellen, bietet das  message-Widget: 

message message ?optionen? 

Es ist vor allem für Dialogboxen gedacht. Bei der Konfiguration einer message gibt man entweder mit 

-width pixel 

eine feste Breite an oder man definiert das Verhältnis von Breite und Höhe: 

-aspect aspect    . 

aspect ist hierbei 100 * Breite / Höhe. Das Default ist 150. 

Wenn der auszugebende Text länger als eine Zeile ist, wird er passend umgebrochen.   

Beispiel

label .status -textvariable status -relief sunken -borderwidth 4 -pady 3 -padx 3

message .msg -textvariable status -aspect 100 -relief sunken -borderwidth 4 -pady 3 -padx 3

message .msg2 -textvariable status -aspect 200 -relief sunken -borderwidth 4 -pady 3 -padx 3

pack .status .msg .msg2 -side left -padx 10

set status "Links ein Label-Widget Rechts zwei Message-Widgets mit aspect \ 100 und 200" 


message-Optionen

Die übrigen Optionen für Messages entsprechen denen des label-Widgets. 

-anchor  center|n|ne|e|se|s|sw|w|nw 
-background (-bg)  farbe 
-borderwidth   width 
-cursor  cursor 
-font  font 
-foreground (-fg)  farbe 
-highlightbackground  color 
-highlightcolor  color 
-highlightthickness  pixel 
-justify   center|right|left 
-padx breite 
-pady höhe 
-relief  flat|raised|sunken|groove|ridge 
-takefocus   0|1 
-text  text 
-textvariable  var 

Man beachte das Fehlen von -height und  -wraplength. 


Checkbuttons

Checkbuttons dienen dazu, Ja/Nein-Optionen anzugeben. Sie unterscheiden sich 
von gewöhnlichen Buttons durch den zusätzlichen Indikator. 

Zu einem Checkbutton gehört normalerweise eine Variable, in der der Zustand des Checkbuttons gespeichert wird. 

checkbutton checkbutton ?-variable var? 

Checkbuttons rufen meistens keine direkten Callbacks auf. Stattdessen wird ihr 
Zustand vom Programm aus abgefragt. 
  

Beispiel

checkbutton .c_fett -variable fett   -text Fett 

checkbutton .c_kurs -variable kursiv -text Kursiv 

checkbutton .c_unte -variable unter  -text Unterstrichen 

button .schriftart -text Schriftart -command schriftart

pack .schriftart -side bottom -anchor e

pack .c_fett .c_kurs .c_unte -side left -pady 5 -padx 10

proc schriftart {} {

        global fett kursiv unter 

        set Typ ""

        if { $fett } { append Typ " bold" }

        if { $kursiv } { append Typ " italic" }

        if { $unter } { append Typ " underline" }

        if { $Typ == "" } { set Typ " normal" }

        puts $Typ

}

Der Wert eines Checkbuttons

Mit der Option 
-variable var 

wird eine Variable angegeben, die den Zustand des Checkbuttons enthält. Sie verhält sich genauso wie die Option -textvariable : 

Eine Änderung des Checkbuttons ändert auch die Variable und eine Änderung der Variablen den Checkbutton. 

Achtung: Falls mit -command ein Callback angegeben wurde, wird dieser durch eine Änderung der Variablen nicht aufgerufen. 

-onvalue wert 

gibt an, welcher Wert der Variablen zugewiesen wird, 

wenn der  Checkbutton angeschaltet ist. Das Default ist 1.  

-offvalue wert      
gibt den Wert für "aus" an. Das Default ist 0. 

Beispiel

checkbutton .c_fett -variable fett   -text Fett     -onvalue bold -offvalue "" 

checkbutton .c_kurs -variable kursiv -text Kursiv    -onvalue italic -offvalue ""

checkbutton .c_unte -variable unter  -text Unterstrichen    -onvalue underline -offvalue ""

button .schriftart -text Schriftart -command schriftart

pack .schriftart -side bottom -anchor e

pack .c_fett .c_kurs .c_unte -side left -pady 5 -padx 10

proc schriftart {} {

        global fett kursiv unter 

        set Typ ""

        append Typ " $fett"

        append Typ " $kursiv"

        append Typ " $unter"

        set Typ [string trim $Typ]

        if { $Typ == "" } { set Typ "normal" }

        puts $Typ

}

Der Indikator

Das zusätzliche grafische Element eines Checkbuttons ist der Indikator. Mit -selectcolor farbe kann man die Farbe des Indikators angeben. 

Wenn man den Indikator nicht anzeigen will, kann man dies mit -indicatoron 0  erreichen. In diesem Fall wird der Hintergrund des Checkbuttons in der selectcolor angezeigt, wenn der Checkbutton aktiv ist. 

Wenn man statt eines Textes mit -image ein Bild auf dem Checkbutton anzeigt, gibt 
        -selectimage  image 
das Bild an, das gezeigt wird, wenn der Button aktiv ist. 

Beispiel

checkbutton .c1 -text "Mit Indikator" -selectcolor green  -foreground blue -background white

checkbutton .c2 -text "Ohne Indikator" -selectcolor green  -indicatoron 0 -foreground blue \

-background white

pack .c1 .c2 -side left -padx 10 -pady 10

Checkbuttons mit Callbacks

Mit der Option -command command kann genau wie bei einem normalen Button ein Befehl angegeben werden, der aufgerufen wird, wenn der Checkbutton an- oder ausgeschaltet wird.  

Beispiel

button .a -text Alpha 

button .b -text Beta

button .q -text Quit -command exit 

checkbutton .d -text "Schalter anzeigen" -variable EnDis  -command { endisable .a .b .q }

set EnDis 1

pack .d -pady 5

pack .a .b .q -side left

update idletasks

pack propagate . false

proc endisable { args } {

        global EnDis

        if { $EnDis } {

                set cmd "pack $args -side left"

        } else {

                set cmd "pack forget $args"

        }

        eval $cmd

}

Weitere Optionen

Checkbuttons können genauso wie normale Buttons konfiguriert werden. Die entsprechenden Optionen werden daher nicht noch einmal besprochen. 

-activebackground   farbe 

-activeforeground   farbe 

-anchor  center|n|ne|e|se|s|sw|w|nw 

-background (-bg)  farbe 

-bitmap bitmap 

-borderwidth   width 

-cursor  cursor 

-disabledforeground  color 

-font  font 

-foreground (-fg)  farbe 

-height  höhe 

-highlightbackground  color 

-highlightcolor  color 

-highlightthickness  pixel 

-image image 

-justify   center|right|left 

-padx breite 

-pady höhe 

-relief  flat|raised|sunken|groove|ridge 

-state   normal|disabled 

-takefocus   1|0 

-text  text 

-textvariable  var 

-underline  index 

-width  breite 

-wraplength  n   

Checkbuttons programmieren

Neben den üblichen Befehlen configure,cget und flash gibt es noch weitere Befehle, um vom Programm aus einen Checkbutton zu bedienen:   

$checkbutton select      
schaltet den checkbutton an.        

$checkbutton deselect      
schaltet den checkbutton aus. 

$checkbutton toggle      
schaltet den checkbutton um.   

$checkbutton invoke      
ruft den checkbutton auf, 

 

wie wenn er mit der Maus betätigt worden wäre. 

Nur invoke ruft einen zum Checkbutton gehörigen Callback auf ! 

Beispiel

checkbutton .c -text Checkbutton -variable C -command { puts "Aufgerufen" } -relief raised \

        

-anchor w -highlightthickness 0 -padx 10

pack    .c -fill x -pady 10 -padx 10

button .s -text Select -command { .c select }

button .d -text Deselect -command { .c deselect }

button .t -text Toggle -command { .c toggle }

button .i -text Invoke -command { .c invoke }

pack .s .d .t .i -side left

Radiobuttons

Radiobuttons ähneln Checkbuttons. Sie ermöglichen es, eine Antwort aus mehreren Möglichkeiten auszuwählen. 

Eine Gruppe von zusammengehörigen Radiobuttons wird durch eine gemeinsame Variable verbunden. 

Ein Radiobutton wird durch 

radiobutton radiobutton  ?optionen? 

erzeugt.  

Beispiel

checkbutton .c -text Student -variable stud

radiobutton .rm -text Männlich -variable weib -value 0

radiobutton .rf -text Weiblich -variable weib -value 1

pack .c -anchor w -padx 5 -pady 5

pack .rm .rf -side left -padx 5 -pady 5

Der Wert eines Radiobuttons

Genauso wie beim Checkbutton wird mit 
-variable var 
eine Variable angegeben, die eine Gruppe von Radiobuttons verbindet und den ausgewählten Wert enthält. 

Wenn kein Radiobutton ausgewählt wurde und der Variablen kein Wert zugewiesen wurde, hat sie den Wert "". Wenn der Variablen ein Wert zugewiesen wird, der zu keinem Radiobutton gehört, ist kein Button ausgewählt. 

Im Gegensatz zu Checkbuttons, die mit -onvalue und -offvalue nur zwei Werte kennen und einen Defaultwert haben, muss jedem Radiobutton mit -value wert ein Wert zugewiesen werden. 

Beispiel

frame .bg

label .bg.l -text Hintergrund

pack .bg.l -side left -padx 10 

foreach color { blue green yellow red black white } {

        radiobutton .bg.$color -text $color \

                -variable BG -value $color

        pack .bg.$color -side left

}

pack .bg -side top -pady 5

set BG cyan

frame .fg

label .fg.l -text Vordergrund

pack .fg.l -side left -padx 10 

foreach color { blue green yellow red black white } {

        radiobutton .fg.$color -text $color \

                -variable FG -value $color

        pack .fg.$color -side left

}

pack .fg -side top -pady 5

set FG black

button .q -text Quit -command exit

button .c -text Config -command { .c config -fg $FG -bg $BG }

pack .q .c -side right -pady 5

pack .c -fill x -expand true

Radiobutton-Optionen

Die übrigen Optionen für einen Radiobutton sind dieselben wie für einen Checkbutton: 

-activebackground   farbe 

-activeforeground   farbe 

-anchor  center|n|ne|e|se|s|sw|w|nw 

-background (-bg)  farbe 

-bitmap bitmap 

-borderwidth   width 

-command  befehl 

-cursor  cursor 

-disabledforeground  color 

-font  font 

-foreground (-fg)  farbe 

-height  höhe 

-highlightbackground  color 

-highlightcolor  color 

-highlightthickness  pixel 

-image image 

-indicatoron  1 | 0 

-justify   center|right|left 

-padx breite 

-pady höhe 

-relief  flat|raised|sunken|groove|ridge 

-selectcolor   farbe 

-selectimage image 

-state   normal|disabled 

-takefocus   1|0 

-text  text 

-textvariable  var 

-underline  index 

-width  breite 

-wraplength  n   



Radiobuttons programmieren

Auch die Programmierung eines Radiobuttons ähnelt der eines Checkbuttons. Es fehlt nur der Befehl toggle.   

$radiobutton configure      
konfiguriert den radiobutton.        

$radiobutton cget option 
liest den Wert einer Option.       

$radiobutton flash      

lässt den radiobutton blinken.      

$radiobutton select 

wählt den radiobutton aus.      

$radiobutton deselect      
schaltet den radiobutton aus und setzt die Variable auf  "". 

$radiobutton invoke      
ruft den radiobutton und ggf. den Callback auf.  

Skalen

Das scale-Widget erzeugt eine Skala, auf der ein numerischer Wert dargestellt wird. 

scale skala ?optionen? 

Mit   -orient  vertical | horizontal 
wird angegeben, wie die Skala dargestellt werden soll. 

Der Wert wird entweder über eine Variable 
  -variable  var 
manipuliert oder direkt mit 

$skala get   und   $skala set    gesetzt und gelesen. 

Beispiel

scale .s1 -orient horizontal -bg white 

pack .s1

.s1 set 23

Skalen beschriften

Das scale-Widget hat zwei spezielle Elemente, den Slider und den darunter liegenden Trough. Außerdem gibt es einige Möglichkeiten um die Skala zu beschriften. 

-sliderrelief  raised|flat|sunken|groove|ridge      ändert das Relief des Sliders.   

-sliderlength bildschirmeinheiten 
gibt die Breite des Sliders an.        

-troughcolor farbe      

ist die Hintergrundfarbe des Sliders.        

-label text 

     

beschriftet die Skala mit text.        

-tickintervall  interval 
zeigt alle interval Werte an der Skala an. 0 bedeutet keine  Anzeige.       

-showval  1|0      
gibt an, ob der Wert angezeigt wird.        

-digits digits      
gibt die Anzahl der Stellen von Wert und Intervall an. 

Beispiel

scale .s1 -orient horizontal -length 15c  -bg white -fg blue -troughcolor lightblue \

        -tickinterval 20 -label "Aufnahmebereitschaft (Prozent)"

pack .s1

Auflösung und Bereich

Den Wertebereich einer Skala gibt man mit 

-from wert  
 und 

-to wert 

an. Default ist  -from 0 -to 100. 

Mit 
-resolution  res 
legt man die Auflösung der Skala fest. Wenn der Wert im Programm (z.B. über eine Variable) gesetzt wird, wird der auf der Skala gezeigte Wert an die Auflösung angepasst. 

Beispiel

scale .s1 -orient horizontal -length 15c \

        -bg white -fg blue -troughcolor lightblue \

        -from -1.0 -to 1.0 -tickinterval 0.25 \

        -resolution 0.05 -variable W1

pack .s1 

set W1 0.72

Werte direkt verändern

Das scale-Widget bietet auch die Möglichkeit, den Wert einer Skala von der grafischen Oberfläche aus zu verändern. Dazu kann man entweder den Slider mit gedrückter Maustaste 1 ziehen oder durch Mausklicks oder Cursortasten den Slider bewegen und damit den Wert ändern.  

  <Button1>                       
bewegt den Slider zu der Seite, auf der die Maus ist.

  <Button2>                       
setzt den Slider an die Stelle, an der die Maus ist.

  <Up> 

  <Left>                       
erhöht den Wert um resolution

  <Down> 

  <Right>                       
senkt den Wert um resolution

  <Control-Up> 

  <Control-Left>                  
erhöht den Wert um bigincrement

  <Control-Down> 

  <Control-Right>                 
senkt den Wert um bigincrement

  <Home>                       
setzt den Silder an den Anfang der Skala.

  <End>                       
setzt den Slider ans Ende der Skala.

Das Verhalten des Widgets kann dabei mit folgenden Optionen beeinflusst werden: 

-bigincrement  bigincrement 
gibt an, um wieviel der Wert bei großen Änderungen 

geändert wird. Default ist 1/10 der Skalenlänge. 

-repeatdelay millisekunden 
ist die Wartezeit, bis Autorepeat bei gedrückter Maustaste 

gestartet wird. Default sind 300 ms. 

-repeatinterval millisekunden  ist die Zeit zwischen zwei Autorepeats. ( Default 100 ms ) 

Beispiel

scale .s1 -length 15c \

        -bg white -troughcolor lightblue \

        -from 10000 -to 0 -tickinterval 1000 \

        -resolution 100 -bigincrement 500 \

        -repeatdelay 2000 -repeatinterval 1000

pack .s1

Weitere Optionen

Neben den besprochenen Optionen kennt das scale-Widget folgende Standardoptionen: 

-activebackground   farbe 

-background (-bg)  farbe 

-borderwidth   width 

-command  befehl 

-cursor  cursor 

-font  font 

-foreground (-fg)  farbe 

-highlightbackground  color 

-highlightcolor  color 

-highlightthickness  pixel 

-relief  flat|raised|sunken|groove|ridge 

-state   normal|disabled 

-takefocus   1|0 

-width  breite

grid

Der Geometrie-Manager grid hat einen völlig anderen Ansatz als pack: 

Er ordnet Widgets auf einem Gitter an und erzeugt damit ein Aussehen wie bei einer Tabelle oder einem Spreadsheat. 

Das jeweils größte Widget bestimmt dabei die Größe einer Spalte bzw. Zeile.   

Beispiel

message .m -textvariable msg -aspect 300 

button  .q -text Quit  -command exit

button  .c -text Clear -command { set msg "?" }

button  .t -text Msg   -command { set msg $Txt }

set msg ?

grid .m .q

grid .c .t

Widgets anordnen

Die einfachste Art, ein Gitter zu erzeugen, besteht darin, die einzelnen Zeilen anzugeben: 

grid  widget1  ?widget2? ... ?optionen? 

Jeder neue grid-Befehl definiert dabei eine neue Zeile. In der Zeile werden die Widgets in der angegebenen Reihenfolge angeordnet.   

Außer Widgets kann man auch folgende Sonderzeichen angeben:   

-      

erweitert das davorstehende Widget nach rechts. 

^ 

erweitert das darüberstehende Widget nach unten. 

x 

erzeugt ein leeres Feld.       

Beispiel      

	label .l1 -text "Zwei Spalten" -bg lightgreen 

message .m1 -aspect 80 -text "Zwei Zeilen" -bg lemonchiffon 

label .l2 -text "Eine Spalte" -bg lightblue 

button .b2 -text Ende -command exit -bg white 

label .l3 -text Ein -bg cyan 

label .l4 -text Loch -bg cyan 

grid .l1  -  .m1 -sticky nsew

grid .l2 .b2  ^  -sticky nsew

grid .l3  x  .l4 -sticky nsew

Widgets einzeln anordnen

Die oben beschriebene Methode zur Erzeugung eines Gitters ist zwar einfach, aber auch unflexibel. Mehr Konfigurationsmöglichkeiten hat man, wenn man für jedes Widget explizit Zeile und Spalte angibt: 

grid  widget  -row zeile -column spalte ?optionen? 

Zeilen und Spalten werden dabei mit 0 beginnend gezählt. Widgets, die sich über mehrere Zeilen oder Spalten ausdehnen,  konfiguriert man mit den Optionen 

-columnspan spalten  und -rowspan  zeilen. 

Beispiel

Das Beispiel der vorigen Seite erzeugt man auch mit: 

grid .l1 -row 0 -column 0 -columnspan 2 

grid .m1 -row 0 -column 2 -rowspan 2    

grid .l2 -row 1 -column 0 

grid .b2 -row 1 -column 1 

grid .l3 -row 2 -column 0 

grid .l4 -row 2 -column 2

Zellen ausfüllen

Der pack-Manager hat die Optionen -fill und -anchor, um festzulegen, was geschehen soll, wenn ein Widget sein Rechteck nicht ausfüllt. grid verwendet   -sticky  seiten   um anzugeben, an welchen Seiten seiner Zelle ein Widget "feststeckt". Als Seiten gibt man entweder die leere Liste {} an  (dies entspricht -anchor center), oder eine Kombination aus n, s, e und  w. 

Die Angabe von zwei gegenüberliegenden Seiten (ns oder ew) bewirkt dabei ein Füllen der Zelle in der entsprechenden Richtung. 

 Beispiel 

frame .r1 -height 15m -width 3 -bd 2 -relief sunken

frame .r2 -height 15m -width 3 -bd 2 -relief sunken

frame .r3 -height 15m -width 3 -bd 2 -relief sunken

frame .c1 -width 25m -height 3 -bd 2 -relief sunken

frame .c2 -width 25m -height 3 -bd 2 -relief sunken

grid .r1 -row 0 -column 1 -sticky nsew

grid .r2 -row 1 -column 1 -sticky nsew

grid .r3 -row 3 -column 1 -sticky nsew

grid .c1 -row 2 -column 0 -sticky nsew

grid .c2 -row 2 -column 2 -sticky nsew

label .l1 -text nichts

label .l2 -text nw

label .l3 -text ew

label .l4 -text news

label .l5 -text ns

label .l6 -text s

grid .l1 -row 0 -column 0

grid .l2 -row 0 -column 2 -sticky nw

grid .l3 -row 1 -column 0 -sticky ew

grid .l4 -row 1 -column 2 -sticky news

grid .l5 -row 3 -column 0 -sticky ns

grid .l6 -row 3 -column 2 -sticky s

Padding

Genau wie pack verfügt grid über Optionen, um einem Widget mehr Platz zu verschaffen:   

-ipadx breite 

-ipady höhe      
fügen zusätzlichen Platz im Widget hinzu.        

-padx breite 

-pady höhe 

fügen zusätzlichen Platz zwischen Widget und Zelle ein. 

Beispiel      

	button .b1 -text Button

button .b2 -text Button

button .b3 -text Button -bg lightblue

button .b4 -text Button -bg lightblue

button .b5 -text Button

button .b6 -text Button

grid .b1 .b2 -sticky news -padx 10  -pady 10

grid .b3 .b4 -sticky news

grid .b5 .b6 -sticky news -ipadx 10 -ipady 10

Größenänderungen

Mit den Befehlen 

grid columnconfigure master  spalte  -weight n    und 

grid rowconfigure       master   zeile   -weight  n 

legt man fest, wie ein Gitter auf Änderungen der Fenstergröße reagieren soll: 

Zusätzlicher Platz wird nach dem Verhältnis der Gewichte verteilt, d.h., eine 

Zeile mit dem Gewicht 2 wächst doppelt so schnell wie eine mit dem Gewicht 1. 

Eine Gewichtung mit 0 bewirkt, dass eine Zeile bzw. Spalte nicht wächst. 

Bei Verkleinerung des Fensters wirken die Gewichte genauso. 

master 


ist der Name des Gitters. 

spalte bzw.  zeile 
kann auch eine Liste von Indizes sein. 

Mit   grid propagate  master  true|false 
legt man wie bei pack fest, 

ob ein Gitter seine Größe ändern darf.  

Beispiel     

array set Farbe { 0 white 1 aquamarine 2 peachpuff  3 paleturquoise 4 lightyellow }

set NF [ array size Farbe ]

set NR 3                            ;# Zeilenzahl

set NC 4                            ;# Spaltenzahl

set NN [ expr ($NR + +1) * $NC ]    ;# Größe des Gitters

label .headline -text "Unterschiedliche Gewichte" -bg paleturquoise  -padx 5 -pady 5

grid .headline -row 0 -column 0 -columnspan $NC -sticky news

for { set i $NC } { $i < $NN } { incr i } {

     button .b$i -text "Button $i" -bg $Farbe([expr $i % $NF]) -padx 5 -pady 5 -anchor w

     grid .b$i -row [expr $i / $NC] -column [expr $i % $NC] -sticky nswe

}

grid columnconfigure . 0 -weight  4

grid columnconfigure . { 1 2 } –weight


Spalten und Zeilen konfigurieren

Neben der Verteilung von Gewichten gibt es noch zwei weitere Optionen für grid columnconfigure und grid rowconfigure:   

-minsize pixel      gibt die minimale natürliche Größe an.        

-pad pixel      gibt ein Padding für die Zeile bzw. Spalte an.       

grid columnconfigure master  spalte  

und 

grid rowconfigure       master   zeile 

ohne Option liefern die Konfiguration der Zeile bzw. Spalte. 

Beispiel  

# Beispiel 6 mit:

grid columnconfigure . 2 -pad 8m

grid rowconfigure . 0 -minsize 19m

Widgets entfernen

Um ein Widget aus einem Gitter zu entfernen, gibt es zwei Befehle:   

grid forget widget  ?widget2? ...      entfernt Widgets und vergisst ihre Konfiguration.        

grid remove widget  ?widget2? ...    entfernt ebenfalls Widgets, merkt sich aber ihre Konfiguration. 

Beispiel                               

 nach  remove/grid              


 nach  forget/grid

button .forget -text forget -command { grid forget .quit }

button .remove -text remove -command { grid remove .quit }

button .quit -text quit -command exit

button .grid -text grid -command { grid .quit }

grid .forget .remove -sticky news -ipadx 10 -ipady 10

grid .quit .grid -sticky news -ipadx 10 -ipady 10

Informationen von grid

Auch der Geometrie-Manager grid ermöglicht es, Informationen über ein Gitter und Widgets zu erfragen.   

grid bbox master ?spalte1 zeile1? ?spalte2 zeile2? 

liefert eine Liste { xoffset yoffset width height }, die den Abstand     zur linken oberen Ecke und die Größe in Bildschirmeinheiten für 

     - das gesamte Gitter (keine Spalte und Zeile), 

     - eine Zelle im Gitter, 

     - die Zellen zwischen {spalte1 zeile1} und {spalte2 zeile2}     angibt.        

grid info widget      

liefert die Konfiguration eines Widgets.        

grid location master x y     
liefert Spalte und Zeile der Zelle, die der relativen Position x,y     am nächsten ist.        

grid size master      

liefert die Größe der Gitters als Liste { spalten zeilen }.        

grid slaves master ?-row zeile? ?-column spalte?      liefert eine Liste der Widgets in master. 

pack und grid

Die beiden Geometrie-Manager pack und grid haben jeweils ihre Stärken und Schwächen. Es ist daher häufig sinnvoll, beide für eine grafische Oberfläche zu verwenden. 

Ein Parent-Widget ( root-window, frame, toplevel ) kann nur von einem Geometrie-Manager verwaltet werden. Verschiedene Frames dürfen verschiedene Geometrie-Manager haben. 

Beispiel  (Programmfragment)

set ncol 6              ;# Spaltenzahl 

frame  .buttons 

button .buttons.quit -text Quit    -command exit

button .buttons.cmd  -text Command -command Cmd 

button .buttons.non  -text None    -command None

button .buttons.all  -text All     -command All

button .buttons.sh   -text Shell   -command Shell

pack .buttons.non .buttons.all -side left -padx 2

pack .buttons.quit             -side right -padx 20

pack .buttons.cmd .buttons.sh  -side right -padx 2

pack .buttons -fill x -pady 4  -side bottom

frame .hosts -relief ridge -borderwidth 2

set i 0

foreach name $HostList {

    checkbutton .hosts.c$name -text "$name"

    grid .hosts.c$name -row [expr $i / $ncol] -column [expr $i % $ncol]

    incr i

}

pack .hosts -fill x -pady 4 -side bottom -ipady 4

frame  .cmd

label  .cmd.label -text Cmd: 

entry  .cmd.entry -background white -foreground black

button .cmd.clear -text Clear -command { .cmd.entry delete 0 end }

pack .cmd.label -side left -anchor w -padx 5

pack .cmd.entry -side left -fill x -expand yes

pack .cmd.clear -side right -anchor e -padx 4

pack .cmd -fill x -pady 4 -side top

Das Entry-Widget

Mit den bisher besprochenen Elementen von Tk können nur Mausklicks verarbeitet werden. Das Entry-Widget dient dazu, Texteingaben des Benutzers zu lesen: 

entry entry   ? -optionen? 

In den meisten Anwendungen wird ein Label vor die Eingabefelder gesetzt, das den Inhalt beschreibt. 

Beispiel

	label .l1 -text "Name" -anchor w

entry .e1

label .l2 -text "Vorname" -anchor w

entry .e2 

grid .l1 .e1 -padx 5 -pady 2 -sticky ew

grid .l2 .e2 -padx 5 -pady 2 -sticky ew

Texteingaben

Auf den Inhalt eines Entry-Widgets kann auf zwei Arten zugegriffen werden: 

$entry get    liefert als Ergebnis die Eingabe. Wenn auf das Eingabefeld häufiger zugegriffen wird, kann man mit der Option    -textvariable  var 

eine Variable definieren, die mit dem Inhalt des Feldes korrespondiert.  Vor der Verwendung einer Eingabe sollte überprüft werden, ob diese sinnvoll ist.  Eine verdeckte Texteingabe erreicht man, indem man mit der Option  -show  zeichen   ein Zeichen angibt, das anstelle der wirklichen Eingabe angezeigt wird.  

Beispiel

	label .l1 -text "Benutzerkennung" -anchor w

entry .e1 -textvariable user

 label .l2 -text "Kennwort" -anchor w

entry .e2 -textvariable password -show * 

set user grote

set password denkste! 

grid .l1 .e1 -padx 5 -pady 2 -sticky ew

grid .l2 .e2 -padx 5 -pady 2 -sticky ew 

button .tx -text Vars -command { puts "$user $password" }

button .ge -text Get  -command {

 set u [ .e1 get ]

        
set p [ .e2 get ] ; puts "$u $p" }

grid .tx .ge -pady 5
	[image: image37.png][=] grids =]
Button Button

Button

Button

Button

Button







Standardoptionen

Die folgenden Optionen für Entries werden nicht noch einmal besprochen: 

-background (-bg)  farbe 

-borderwidth   width 

-cursor  cursor 

-font  font 

-foreground (-fg)  farbe 

-highlightbackground  color 

-highlightcolor  color 

-highlightthickness  pixel 

-justify   center|right|left 

-relief flat|raised|sunken|groove|ridge 

-state   normal|disabled 

-takefocus   1|0 

-width  breite

Indizes

Um Teile eines Eingabefeldes adressieren zu können, werden die folgenden Indizes verwendet:   

zahl      
Das Zeichen an der Position zahl. Das erste Zeichen hat den      Index 0.        

end      
Die Position hinter dem letzten Zeichen.        

insert     
 Das Zeichen direkt hinter dem Eingabe-Cursor (s.u.).        

sel.first 

sel.last      
Das erste bzw. letzte selektierte Zeichen (s.u.).     

Der Befehl 
$entry  index  index 

liefert den numerischen Wert zu einem Index. 

Beispiel

	entry .e1

grid .e1 - -padx 5 -pady 2 -sticky ew 

button .a -text Prepend -command { .e1 insert 0 A }

button .b -text Append  -command { 

           .e1 insert end [ .e1 index insert ] }\

grid .a .b -pady 5

Der Eingabe-Cursor

markiert die Stelle, an der Eingaben erfolgen. Sein Aussehen kann durch die folgenden Optionen verändert werden:   

-insertbackground farbe      

Farbe des Cursors        

-insertwidth pixel       

Breite des Cursors        

-insertborderwidth pixel       

Größe des Randes für 3D-Effekte        

-insertontime millisekunden 

-insertofftime millisekunden      
Zeit, die der Cursor beim Blinken an bzw. aus ist. 

-insertofftime 0 


verhindert das Blinken. 

Mit 
$entry icursor index 
kann der Cursor an die Stelle index positioniert werden. 

Beispiel

label .l1 
-text "Benutzerkennung" -anchor w

entry .e1 
-textvariable user -bg white -insertbackground lightgreen -insertborderwidth 10 \

-insertwidth 10  -insertontime 8000 -insertofftime 1000 

set user grote 

grid .l1 .e1 -padx 5 -pady 2 -sticky ew 

.e1 icursor 3

Text selektieren

Mit der Maus kann ein Bereich des Eingabefeldes selektiert werden, um zum Beispiel gelöscht oder kopiert zu werden. Das Aussehen und Verhalten des selektierten Textes wird durch folgende Optionen gesteuert:   

-selectforeground farbe     
Vordergrundfarbe        

-selectbackground farbe       
Hintergrundfarbe        

-selectborderwidth pixel       
Randbreite        

-exportselection 1 | 0      
erlaubt das Kopieren des selektierten Textes in andere X11-Programme  bzw. verbietet es. 

Beispiel

label .l1 -text "Name" -anchor w

entry .e1 -textvariable user -bg white   -selectforeground blue  -selectbackground lightgreen 

set user "Mathias Grote" 

grid .l1 .e1 -padx 5 -pady 2 -sticky ew

Mit selektiertem Text arbeiten

Eine Reihe von Befehlen dient dazu, mit der Selektion zu arbeiten:   

$entry selection clear      
bewirkt, dass nichts mehr selektiert ist.   

$entry selection present      
liefert true, wenn etwas selektiert ist, und ansonsten false        

$entry selection range start ende  
setzt den selektierten Bereich von start bis ende (ausschließlich)   

$entry selection adjust index  erweitert den selektierten Bereich bis index (einschließlich)

$entry selection from start
setzt den Anfang des selektierten Bereichs   

$entry selection to ende      
setzt das Ende des selektierten Bereichs   

Beispiel

label .l1 -text "Name" -anchor w

entry .e1 -textvariable user -bg white  -selectforeground blue  -selectbackground lightgreen 

set user "Mathias Grote" 

grid .l1 .e1 - - -padx 5 -pady 2 -sticky ew

button .c -text clear -command { .e1 selection clear }

button .e -text end -command { .e1 selection adjust end }

button .r -text range -command { .e1 selection range 4 6 }

button .f -text from-to  -command { .e1 selection from 2 ;.e1 selection to 7 } 

grid .c .e .r .f -padx 5 -pady 2 -sticky ew

Text löschen und einfügen

Der Befehl 
$entry insert index zeichenkette 
fügt eine Zeichenkette in ein Eingabefeld ein. Mit 
$entry delete start ?ende? 

werden die Zeichen von start bis ende (ausschließlich) gelöscht. Wenn nur ein Index angegeben wird, wird nur das Zeichen an der angegebenen Stelle gelöscht.   

Beispiel  

entry .e1 -textvariable user -bg white    -selectforeground blue  -selectbackground lightgreen 

set user "Mathias Grote" 

grid .e1 - -padx 5 -pady 2 -sticky ew 

button .i -text insert -command { .e1 insert insert x }

button .d -text delete \

    -command { if [.e1 selection present] {

                 .e1 delete sel.first sel.last }

             } 

grid .i .d -padx 5 -pady 2 -sticky ew

Scrollbar-Befehle

Neben den Standardbefehlen configure und cget gibt es beim Entry-Widget noch den Befehl xview, der zur Verknüpfung mit einem Scrollbar dient. 

Für die direkte Programmierung nützlich ist der Unterbefehl   $entry xview moveto 1.0 ,   der das Ende eines Eingabefeldes sichtbar macht. 

Der Scrollbar selbst wird mit der Option -xscrollcommand  definiert. 

Anmerkung: Bei einem Eingabefeld ist ein Scrollbar nicht besonders sinnvoll, da er relativ viel Platz benötigt und man das Eingabefeld genauso schnell mit den Cursortasten scrollen kann.   

Beispiel

[image: image41.png]


  [image: image42.png]- e

J Nane Ribbeck in Havelland





label .l1 -text "Name" -anchor w

entry .e1 -textvariable user -bg white 

set user "Herr von Ribbeck auf Ribbeck im Havelland" 

grid .l1 .e1 -padx 5 -pady 2 -sticky ew 

button .s -text show -command { puts [.e1 xview] }

button .e -text end  -command { .e1 xview moveto 1.0 } 

grid .s .e -padx 5 -pady 2 -sticky ew

Listboxen

Checkbuttons und Radiobuttons bieten die Möglichkeit, aus einer Liste von Optionen auszuwählen. Wenn diese Liste zu umfangreich wird, verwendet man besser eine Listbox: 

	listbox $listbox ?optionen?

Diese zeigt ihren Inhalt zeilenweise an und erlaubt es, auf verschiedene Arten Elemente auszuwählen.  Mit dem Befehl  

$listbox insert index zeile ?zeile ...? 

vor der Stelle index eingefügt

Beispiel

set Colors {

black blue brown cyan gray green magenta orange pink

purple red tomato turquoise violet wheat white yellow 

} 

listbox .color -bg white

pack .color -fill both -expand true 

foreach line $Colors { .color insert end $line }

Optionen für Listboxen

-height  höhe 

-width  breite      
geben die Höhe und Breite der Listbox in Zeichen an. Ein Wert < 1 bewirkt, dass die jeweils benötige Größe  verwendet wird.        

-setgrid 0 | 1      
1 erzwingt, dass beim Verkleinern und Vergrößern der Listbox jeweils ganze Zeilen dargestellt werden.       

-listvariable varname   gibt analog zur Option -textvariable bei anderen Widgets  den Namen einer Liste an, die in der Listbox angezeigt wird.        

Beispiel

set Colors { black white }

listbox .color -bg white -listvariable Colors    -height -1 

 pack .color -fill both -expand true

update 

after 3000

set Colors {

black blue brown cyan gray green magenta orange pink

purple red tomato turquoise violet wheat white yellow }

Standardoptionen für Listboxen

Die weiteren Optionen 

	-background (-bg)  farbe 

-borderwidth   width 

-cursor  cursor 

-exportselection   1|0

-font  font
	-foreground (-fg)  farbe 

-highlightbackground  color 

-highlightcolor  color 

-highlightthickness  pixel 

-relief flat|raised|sunken|groove|ridge 

-selectbackground   farbe
	-selectborderwidth   width 

-selectforeground   farbe 

-takefocus   1|0 

-xscrollcommand  befehl 

-yscrollcommand  befehl


 wirken wie bei anderen Widgets auch. 

Indizes

Die meisten Befehle zur Programmierung von Listboxen verwenden einen Index:   

zahl      
Die Zeile zahl. Die erste Zeile hat die Nummer 0.        

end      
Die letzte Zeile bzw. die Position hinter der letzten  Zeile.        

active      
Das "aktive" Element. Wenn die Listbox den Focus hat, wird es unterstrichen.        

anchor  

Der Ankerpunkt der Selektion. 

Mit 
$lb index index 
erhält man den numerischen Wert zu einem Index. 

  Beispiel

# Ergänzung zu Beispiel 1

proc Show {} {

    global Outaa

    set Outaa "Active: "

    append Outaa [ .color index active ]

    append Outaa "    Anchor: "

    append Outaa [ .color index anchor ]

}

Elemente konfigurieren

Das Aussehen einzelner Zeilen einer Listbox lässt sich mit dem 

Befehl 

$lb itemconfigure index ?option? ?wert? ?option wert ...?   ändern und abfragen. 

Gültige Optionen sind 

-background (-bg)  farbe 

-foreground (-fg)   farbe 

-selectbackground  farbe 

-selectforeground   farbe . 

Zur Abfrage einer einzelnen Option dient 

$lb itemcget index option . 

Beide Befehle funktionieren analog zu configure und cget. 

Beispiel

	set Colors { black blue brown cyan gray green magenta orange \ pink purple red tomato turquoise violet wheat white \ yellow 

} 

listbox .color -bg white -height -1

pack .color -fill both -expand true 

foreach line $Colors {

        .color insert end $line

        .color itemconfig end -fg $line -selectbackground $line

}




Einfügen, Löschen, Lesen

Die folgenden Befehle werden beim Arbeiten mit Listboxen häufig benötigt:   

$lb insert index zeile ?zeile ...?      
fügt Zeilen vor dem Index index ein.        

$lb delete anfang ?ende?      

löscht die Zeilen von anfang bis ende (einschließlich). Wenn ende nicht angegeben wird, 

wird nur eine Zeile gelöscht.        

$lb get anfang ?ende?      

gibt eine Liste der Zeilen zwischen anfang und ende  (bzw. die Zeile anfang) zurück.        

$lb curselection      


liefert eine Liste der selektierten Indizes.    

	Beispiel  

set Colors {

black blue brown cyan gray green magenta orange pink

purple red tomato turquoise violet wheat white yellow } 

listbox .color -bg white -listvariable Colors -height -1 

pack .color -fill both -expand true

.color delete 0 2


Elemente auswählen

Es gibt verschiedene Varianten, um aus einer Listbox Elemente auszuwählen. Die aktuelle Variante wird mit der Option 
-selectmode  browse | single | multiple | extended festgelegt. 

Die Auswahl kann mit folgenden Befehlen bearbeitet werden: 

$lb selection anchor index      
setzt den Ankerpunkt.        

$lb selection clear anfang ?ende?      
entfernt die Zeilen zwischen anfang und ende aus der      Auswahl.   

$lb selection includes index      
gibt an, ob die Zeile index ausgewählt wurde.   

$lb selection set anfang ?ende?      
wählt alle Zeilen zwischen anfang und ende aus, ohne den Status der Zeilen außerhalb diese Bereiches  zu ändern. 

Beispiel

set Colors {

black blue brown cyan gray green magenta orange pink

purple red tomato turquoise violet wheat white yellow } 

listbox .color -bg white -listvariable Colors  -height -1 -width -1 

label .show -textvariable Out -width 40 -anchor w -relief ridge  -justify left -wraplength 10c 

pack .color -fill both -expand true

pack .show -fill x -expand true -side bottom 

proc Show {} {

    global Out

    foreach ix [ .color curselection ] {

        lappend Out [ .color get $ix ]

    }

} 

.color selection set 4 8

Show

Einfache Selektions-Modi

single

Es kann nur ein Element ausgewählt werden. 

Dies geschieht durch Anklicken mit der Maustaste 1. 

browse

Funktioniert wie der single-Modus, bei festgehaltener Maustaste bewegt sich die Auswahl mit der Maus. Dies ist der Default-Modus. 

multiple

Es können mehrere Zeilen ausgewählt werden. Anklicken mit der Maustaste 1 wechselt den Status einer Zeile.   

Beispiel

set Colors {

black blue brown cyan gray green magenta orange pink

purple red tomato turquoise violet wheat white yellow } 

listbox .color -bg white -listvariable Colors  -height 11 -width -1 

bind .color <<ListboxSelect>> Show

label .showaa -textvariable Outaa -width 40  -anchor w -relief ridge

label .show -textvariable Out -width 40 -anchor w -relief ridge  -justify left -wraplength 10c 

radiobutton .rb -variable mode -command Conf   -text browse -value browse

radiobutton .rs -variable mode -command Conf   -text single -value single

radiobutton .rm -variable mode -command Conf  -text multiple -value multiple

radiobutton .re -variable mode -command Conf   -text extended -value extended

.rb select 

pack .color -fill both -expand true

pack .show .showaa -fill x -expand true -side bottom

pack .rb .rs .rm .re -side left -pady 2 -padx 2 

proc Conf {} {

    global mode

    .color selection clear 0 end

    .color config -selectmode $mode

}

proc Show {} {

    global Out Outaa

    set Outaa [ .color get anchor ]        

    foreach ix [ .color curselection ] {

        lappend Out [ .color get $ix ]

    }

}

Der erweiterte Auswahlmodus

Der extended-Modus erlaubt es, mehrere Elemente auszuwählen. Die Bedienung geschieht wie folgt:     

   <Button-1>         
fängt  eine neue Auswahl an und setzt den Ankerpunkt.

   <Shift-Button-1>   
erweitert die Auswahl vom Ankerpunkt bis zum aktuellen Element. 

   <Ctrl-Button-1>    
setzt den Ankerpunkt neu, ohne die  bisherige Auswahl zu löschen. Der Status  der neu ausgewählten Elemente wird  invertiert.Das Ziehen der gedrückten Maustaste erweitert dabei jeweils den ausgewählten Bereich. 

Weitere Befehle

Neben den Standardbefehlen configure und cget sowie den Scrollbefehlen xview und yview gibt es noch folgende Befehle für Listboxen:   

$lb activate index      
macht das angegebene Element zum aktiven Element.       

$lb see index      
rückt das angegebene Element in den sichtbaren Bereich.       

$lb size      

liefert die Anzahl der Zeilen in der Listbox. 

Beispiel

set Colors {

black blue brown cyan gray green magenta orange pink

purple red tomato turquoise violet wheat white yellow 

} 

listbox .color -bg white

pack .color -fill both -expand true

foreach line $Colors {

        .color insert end $line

        .color itemconfig end -fg $line \

                -selectbackground $line

}

.color see end

Scrollbars

Widgets, die nicht ihren ganzen Inhalt auf dem Bildschirm anzeigen, z.B. Listboxen oder Textwidgets, können zwar durch Festhalten der Maustaste oder die Cursortasten bewegt werden, doch ist es einfacher, sie mit Scrollbars zu versehen.   

scrollbar scrollbar -orientation orientation -command scrollcmd ?optionen? 

Dabei gibt orientation an, ob der Scrollbar horizontal oder vertikal sein soll.   

command verbindet den Scrollbar mit dem Widget, das gescrollt werden soll. 

Die Verbindung vom Widget zum Scrollbar geschieht durch die Widget-Optionen 

-xscrollcommand   { scrollbar  set }     und 

-yscrollcommand   { scrollbar  set } .   

Beispiel

text .txt -xscrollcommand { .xscroll set } -yscrollcommand { .yscroll set }         

scrollbar .yscroll -orient vertical   -command { .txt yview }

scrollbar .xscroll -orient horizontal -command { .txt xview } 

grid .txt .yscroll -sticky nsew

grid .xscroll      -sticky nsew

grid columnconfigure . 0 -weight 1

grid rowconfigure . 0 -weight 1 

set File [ open /etc/passwd r ]

set lines [ read -nonewline $File ]

.txt insert 1.0 $lines

Das Scrollbar-Widget

Ein Scrollbar besteht aus dem Hintergrund (trough), dem Schieber (slider) und den beiden Pfeilen (arrows). 

Diese können mit den Optionen   

-troughcolor farbe      
Farbe des Hintergrundes   

-background farbe      
Farbe der aktiven Elemente (Schieber und Pfeile)   

-activebackground farbe     Farbe der aktiven Elemente, wenn die Maus über ihnen ist        

-activerelief relief      
Relief der aktiven Elemente, wenn die Maus über ihnen ist        

-elementborderwidth pixel  Randbreite der aktiven Elemente 

konfiguriert werden.   

Beispiel                                          

scrollbar .yscroll -orient vertical   -command { .txt yview } -activebackground blue 

-background red -troughcolor yellow

scrollbar .xscroll -orient horizontal -command { .txt xview }   -relief raised -activerelief flat

Scrollbar-Optionen

Zusätzlich zu den soeben besprochenen Optionen gibt es die folgenden Standardoptionen: 

-borderwidth   width 

-cursor  cursor 

-highlightbackground  color 

-highlightcolor  color 

-highlightthickness  pixel 

-relief flat|raised|sunken|groove|ridge 

-takefocus   1|0 

-width  breite 

width gibt dabei die Größe der schmalen Dimension des Scrollbars an.

Scroll-Befehle

Die beim Widget mit -xscrollcommand und -yscrollcommand angegebenen Befehle werden beim Aufruf um zwei Parameter erweitert, die die relative Position der ersten und letzten sichtbaren Zeile bzw. Spalte des Widgets angeben. 

Zum Beispiel bedeutet  0.0  0.5, dass die erste Hälfte des Widgets sichtbar ist. 

Dieses Verfahren ist zwar auf den Befehl $scrollbar set zugeschnitten, kann aber auch für andere Funktionen genutzt werden. 

Beispiel                                               

proc ScrollSet { scrollbar first last } {

        if { $first == 0.0 && $last == 1.0 } {

                grid remove $scrollbar

        } else {

                grid $scrollbar

                $scrollbar set $first $last

        }

}

listbox .color -bg white -yscrollcommand { ScrollSet .yscroll } 

scrollbar .yscroll -orient vertical   -command { .color yview }

grid .color .yscroll -sticky nsew

grid columnconfigure . 0 -weight 1

grid rowconfigure . 0 -weight 1 

set Colors {

        black blue brown cyan gray green magenta orange pink

        purple red tomato turquoise violet wheat white yellow }

foreach line $Colors {

        .color insert end $line

}

Scroll-Befehle (2)

Ähnlich wie die Scroll-Befehle des Widgets wird der beim Scrollbar mit -command angegebene Befehl beim Aufruf so erweitert, wie es der xview- bzw yview-Befehl des Widgets erfordert, z.B. mit 

        moveto 0            oder         scroll -1 pages . 

Auch diese Befehle können umprogrammiert werden. 

Beispiel

2 Listboxen mit 1 Scrollbar 

proc DoScroll { args } {

    foreach lb { .color .color1 } {

        eval "$lb yview $args"

    }

} 

proc ScrollSet { first last } {

    .yscroll set $first $last

    foreach lb { .color .color1 } {

        eval "$lb yview moveto $first"

    }

} 

listbox .color  -bg white -yscrollcommand { ScrollSet }

listbox .color1 -fg white -yscrollcommand { ScrollSet } 

scrollbar .yscroll -orient vertical -command { DoScroll } 

grid .color .yscroll .color1 -sticky nsew 

grid columnconfigure . { 0 2 } -weight 1

grid rowconfigure . 0 -weight 1 

set File [ open Colorlist1 r ]

set lines [ read -nonewline $File ]

regsub -all {\n} $lines { } Colors 

foreach line $Colors {

    .color  insert end $line

    .color  itemconfig end -fg $line -selectbackground $line

    .color1 insert end $line

    .color1 itemconfig end -bg $line -selectforeground $line

}

Events

Mit Tk ist es möglich, Ereignissen (Events) Befehle zuzuordnen. Ereignisse sind z.B. das Drücken einer Taste, das Bewegen der Maus oder das Schließen eines Fensters. Damit kann man das Verhalten einer grafischen Oberfläche leicht an eigene Vorgaben anpassen. 

Beispiel                                     

set Colors {

black blue brown cyan gray green magenta orange pink

purple red tomato turquoise violet wheat white yellow 

}

listbox .color -bg white -listvariable Colors  -height -1 -selectmode single

label .show -textvariable Out -width 40 -anchor w -relief ridge

pack .color -fill both -expand true

pack .show -fill x -expand true -side bottom

bind .color <Button-3> {

    lappend Out [ .color get [ .color curselection ]] }

Der bind-Befehl

Der bind-Befehl dient der Abfrage und Definition der Verbindung von Ereignissen und Befehlen (binding) :   

bind  name     

gibt eine Liste mit den Events zurück, für die ein Binding definiert ist.        

bind  name  event      

liefert den an event gebundenen Befehl.   

bind  name  event   command      
definiert den Befehl, der beim Eintritt des Ereignisses event ausgeführt werden soll.        

bind  name  event   +command      
fügt den Befehl command zu den Bindings für event hinzu. name ist dabei der Name eines Widgets oder einer Widgetklasse, z.B. Listbox, Frame oder  Entry.   

Wie auch bei den durch -command mit Widgets verbundenen Befehlen gilt:   Befehle werden im globalen Adressraum ausgeführt.  

Beispiel

puts [ bind Button ]

foreach binding [ bind Button ] {

    puts "$binding "

    puts "[ bind Button $binding ] \n"

}

Einfache Event-Spezifikationen

Die vollständige Syntax zur Angabe eines Events ist < modifier-modifier-event-detail > .    Sie wird aber nur bei Tastatur- und Maus-Ereignissen benötigt. 

Bei allen anderen Ereignissen wird nur 
< event > 
angegeben:   

<Enter> 

<Leave>     

 Der Mauszeiger betritt bzw. verlässt das Widget.        

<FocusIn> 

<FocusOut>     
 Das Widget erhält bzw. verliert den Tastaturfokus.        

<Map> 

<Unmap>      

Das Widget erscheint auf dem Bildschirm bzw. verschwindet, 



z.B. wird zum Icon.        

<Configure>      
Das Widget wird umkonfiguriert, z.B. verkleinert oder vergrößert.       

Beispiel                            

set i 1

foreach e { Name Vorname Telefon Email Stadt Strasse } {

    entry .ent$i 

    label .lab$i -text " $e: " -anchor w 

    grid .lab$i .ent$i -sticky ew

    incr i

}

set bg [ .ent1 cget -bg ]

bind Entry <FocusIn>  { %W config -bg white }

bind Entry <FocusOut> { %W config -bg $bg }

Tastatur und Maus

Die Tastatur- und Maus-Events sind:   

<ButtonPress> , <Button> 

<ButtonRelease>      
Eine Maustaste wird gedrückt bzw. losgelassen.        

<Motion>      

Die Maus wird über dem Widget bewegt.        

<KeyPress> , <Key> 

<KeyRelease>      
Eine Taste wird gedrückt bzw. losgelassen.        

Mit Ausnahme von <Motion> können diese Ereignisse durch ein Detail näher spezifiziert werden, das angibt, welche Taste gemeint ist. Wird kein Detail angegeben, so löst jede Taste das Ereignis aus. 

Mausdetails

Bei Maus-Ereignissen kann die Nummer einer Maustaste ( 1 - 5 ) angegeben werden. 

Tastaturdetails

Hier wird entweder die entsprechende Taste oder ein symbolischer Name angegeben. Symbolische Namen sind u.a. 

Return, Escape, Tab, Up, Down, Left, Right, BackSpace. 

Bei Tasten ist Groß- und Kleinschreibung signifikant: <Key-A>  bedeutet  dasselbe wie <Shift-Key-a> ! Details können abgekürzt werden, dies sollte man aus Gründen der Lesbarkeit aber nur in eindeutigen Fällen tun. 

Beispiele

<Button>      Drücken eine beliebigen Maustaste

<Button-3>    Maustaste 3

<3>           dasselbe, nicht Taste 3

<KeyPress-a>  Taste a

<Key-a>       
dasselbe

<a>           
auch Taste a

<Control-c>   
hier ist die Abkürzung sinnvoll

<Return>      
hier auch

<Enter>       
ist kein Tastatur-Event !!

Modifier

Spezielle Tastatur- und Maus-Ereignisse können auch dann eintreten, wenn vor dem Event schon ein oder zwei weitere Tasten gedrückt sind. Diese Ereignisse werden durch sogenannte Modifier angegeben:   

Control 

Shift 

Alt 

Meta 

Die entsprechende Taste muss beim Eintreten des Events schon gedrückt sein.        

Button1 , B1 

Button2 , B2 

Button3 , B3 

Button4 , B4 

Button5 , B5 

Die entsprechende Maustaste muss beim Eintreten des Events schon gedrückt sein. Es wird jeweils nur das detaillierteste Ereignis ausgeführt. 

Beispiel                                                       

text .t -bg white

pack .t -fill both -expand true 

bind .t <Button-1> {

        .t insert end "Button 1\n" }

bind .t <Shift-Button-1> {

        .t insert end "Shift-Button 1\n" }

bind .t <Control-Shift-Button-1> {

        .t insert end "Control-Shift-Button 1\n" } 

bind .t <Button2-Button-1> {

        .t insert end "Button2-Button 1\n" }

Mehrfache Mausklicks

Das mehrfache schnelle Betätigen einer Maustaste wird durch die Modifier   

Double      
Doppelklick 

Triple      
Dreifachklick        

beschrieben. 

Das Ereignis <Double-Button-1> unterscheidet sich von <Button-1><Button-1> dadurch, dass die Mausklicks innerhalb einer gewissen Zeit stattfinden müssen. 

Achtung: 
<Triple-1> löst auch <Button-1> und <Double-1> aus! 

Beispiel                                                 

text .t -bg white

pack .t -fill both -expand true 

bind .t <Button-3> {

        .t insert end "Button 3\n" }

bind .t <Double-Button-3> {

        .t insert end "Double-Button 3\n" }

bind .t <Triple-Button-3> {

        .t insert end "Triple-Button 3\n" }

Ereignis-Folgen

Beim bind-Befehl ist es möglich, Befehle für Ereignisse anzugeben, die aus mehreren Einzel-Events bestehen.   

bind name  <event1><event2>....  <command>   

Dies geschieht meistens für Tastatureingaben. 

Auch für Ereignisfolgen gilt wie für mehrfache Mausklicks, dass Befehle, die mit den vorderen Ereignissen verbunden sind, ebenfalls ausgeführt werden.   

Beispiel

# Hintertür für ein Spielprogramm

set Mogelmodus 0

bind . <Key-M><o><g><e><l> { set Mogelmodus 1 }

Automatische Variablen-Substitution

In den von einem Event ausgeführten Befehl können einige Variable beim Aufruf von Tk eingesetzt werden:   

%W      Name des Widgets 

%b      Nummer der Maustaste (nur bei Maus-Events) 

%A 

%K      ASCII-Zeichen und symbolischer Name bei      <KeyPress> und <KeyRelease> 

%h 

%w      Höhe und Breite des Widgets ( bei <Configure> ) 

%x 

%y      Koordinaten des Mauszeigers relativ zur linken oberen Ecke (0,0) des Widgets. 

%X 

%Y      Koordinaten relativ zur linken oberen Ecke des Desktops   

Beispiel                             

# Anzeige von Tasten und Tastensymbolen 

label .msg -width 30 -text "Beliebige Taste drücken"

pack .msg 

bind .msg <Key> {.msg configure -text "keysym = %K \n key = %A"}

focus .msg

Virtuelle Ereignisse

Mit dem event-Befehl ist es möglich, reale Ereignisse einem virtuellen Ereignis zuzuweisen. Dies erleichtert u.a. das Schreiben von plattformunabhängigen Programmen. Virtuelle Ereignisse haben die Form <<virt>> . 

Die virtuellen Ereignisse können anschließend wie reale Ereignisse verwendet werden. 

Die Syntax des Befehls ist   

event add <<virt>> event ?event ...?  fügt reale Ereignisse zu einem virtuellen hinzu.        

event delete <<virt>>      

löscht ein virtuelles Ereignis.        

event info     
 


gibt eine Liste aller definierten virtuellen Ereignisse aus.        

event info <<virt>>      

gibt eine Liste der realen Ereignisse zu <<virt>> aus.       

Beispiel

event add <<Cancel>> <Control-c>

event add <<Cancel>> <Escape> 

bind . <<Cancel>> { exit } 

foreach ev [ event info ] {

        puts "$ev : [ event info $ev ] \n"

}

Der bindtags-Befehl

Ein binding tag fasst eine Gruppe von Verknüpfungen zusammen. Dies ermöglicht es, Funktionalitäten zusammenzufassen und Widgets flexibel zu konfigurieren. Normalerweise gehören zu einem Widget die Tags   

  Widgetname (z.B. .quit ) 

  Widgetklasse (z.B. Button ) 

  Toplevel-Widget (z.B. . ) 

  all   

Bei toplevel-Widgets entfällt der dritte Punkt. Mit dem Befehl   

bindtags widget   

werden die zu einem Widget gehörenden Tags angezeigt. Mit   

bindtags widget taglist   

wird dem Widget eine Liste von Tags zugewiesen.   

Beispiel

# Die Bindings eines Widgets ausgeben

proc bindings { widget } {

        puts "$widget"

        foreach tag [bindtags $widget] {

                puts [format "\t%-20s" $tag]

                foreach spec [bind $tag] {

                        puts [format "\t\t%-20s" $spec]

                        set cmd [bind $tag $spec]

                        set cmd [string trim $cmd "\n"]

                        regsub -all "\n" $cmd "\n\t\t\t" cmd

                        puts [format "\t\t\t%-20s" $cmd]

                }

        }

} 

button .q -text Quit -command exit

bind .q <Return> exit

pack .q 

bindings .q

Mehrfache Verknüpfungen

Wenn für ein Ereignis bei mehreren Tags ein Binding definiert ist, werden alle Verknüpfungen in der Reihenfolge der Tags ausgeführt, im Beispiel für .entry1 erst  das Binding für .entry1, dann das für Entry. 

Die Befehle break und continue haben für Bindings eine spezielle Bedeutung: 

break 

beendet die weitere Bearbeitung des Ereignisses, 

continue 
setzt sie mit dem nächsten Tag fort. 

Beispiel                              

label .label1 -text "Nummer" -anchor w

entry .entry1 -bg white

label .label2 -text "Nummer" -anchor w

entry .entry2 -bg white 

bind .entry1 <KeyPress> {

        if { ![ regexp "\[0-9\]" "%A"] } {

                bell

        }

} 

bind .entry2 <KeyPress> {

        if { ![ regexp "\[0-9\]" "%A"] } {

                bell

                break

        }

} 

grid .label1 .entry1 -padx 5 -pady 2 -sticky ew

grid .label2 .entry2 -padx 5 -pady 2 -sticky ew 

focus .entry1

Eigene Tags

Durch die Angabe eines eigenen Namens für ein Binding kann man Tags definieren, die anschließend mit einem bindtags-Befehl einem Widget zugewiesen werden können. Dies ermöglicht die einfache Konfiguration einer ganzen Gruppe von Widgets.  

Beispiel

bind nummer <KeyPress> {

        if { ![ regexp "\[0-9\]" "%A"] } {

                bell

                break

        }

} 

label .label1 -text "Nummer" -anchor w

entry .entry1 -bg white 

bindtags .entry1 "nummer [ bindtags .entry1 ]" 

grid .label1 .entry1 -padx 5 -pady 2 -sticky ew

focus .entry1

after

Neben Ereignissen, die mit einer Maus- oder Tastatur-Eingabe verbunden sind, kennt Tk auch solche, die zu einem späteren Zeitpunkt stattfinden sollen. Diese werden mit dem after-Befehl verwaltet.   

after ms      

erzeugt keinen Event, sondern macht eine Pause von  ms Millisekunden.        

after ms befehl ?arg ...?     
führt den angegebenen Befehl nach ms Millisekunden aus. 

Das Programm läuft in der Wartezeit weiter. Der after-Befehl gibt einen Identifier zurück.        

 after idle befehl ?arg ...?       führt den angegebenen Befehl aus, wenn keine anderen Events mehr vorhanden sind.        

after cancel id 

after cancel befehl      
löscht den durch den Identifier bzw. Befehl spezifizierten Event aus der Warteschlange.        

after info ?id?      

liefert eine Liste der ausstehenden after-Events bzw. den      Befehl zu dem angegebenen Event. 

Beispiel                                                                

proc AddHelp { widget text } {

    global Help

    set Help($widget) $text

    bind $widget <Enter> "ShowHelp $widget"

    bind $widget <Leave> "CancelHelp"

} 

proc ShowHelp { widget } {

    global Help

    if [info exists Help($widget) ] {

        set Help(Pending) [after 1000 [list set Helptext $Help($widget)]]

    }

}

proc CancelHelp { } {

    global Help Helptext

    if [info exists Help(Pending) ] {

        after cancel $Help(Pending)

        unset Help(Pending)

    }

    set Helptext ""

}

# ......

AddHelp .buttons.non  "Alle Hosts löschen"

AddHelp .buttons.all  "Alle Hosts auswählen"

AddHelp .buttons.cmd  "Befehl ausführen"

AddHelp .buttons.term "Terminalfenster öffnen"

AddHelp .buttons.quit "Anwendung beenden"

Menüs

Ein Menü dient dazu, Buttons, Checkbuttons und Radiobuttons, die nicht permanent benötigt werden, von einer grafischen Oberfläche zu entfernen. Ein Menü wird durch 

menu name ?optionen? 

erzeugt. Im Gegensatz zu anderen Widgets wird es nicht durch einen Geometrie-Manager dargestellt, sondern durch Aufruf eines Menubuttons oder eines Buttons in einem Menubar. Dieser Menubutton bzw. Menubar muss das Parent-Widget des Menüs sein.   

Beispiel

menu .mbar.cmd 

.mbar.cmd add command -label "Show" \

        -command { set Value "$Mbar_col $Mbar_cb1 $Mbar_cb2" }

.mbar.cmd add command -label "Config" \

        -command { .value config -fg $Mbar_fg -bg $Mbar_bg }

.mbar.cmd add separator

.mbar.cmd add command -label "Quit" -command exit -foreground red

Optionen für Menüs

Das Menu-Widget verwendet die Standard-Optionen 

-activebackground   farbe 

-activeforeground   farbe 

-activeborderwidth   width 

-background (-bg)  farbe 

-borderwidth   width 

-cursor  cursor 

-disabledforeground  color 

-font  font 

-foreground (-fg)  farbe 

-relief flat|raised|sunken|groove|ridge 

-selectcolor   farbe 

-takefocus   1|0 

Mit
-postcommand command   kann ein Befehl angegeben werden, der ausgeführt wird, bevor das Menü auf dem Bildschirm dargestellt wird.   

Menü-Elemente

Die Einträge in ein Menü werden am einfachsten durch den Befehl 


$menu add  typ ?optionen? 

angegeben. Sie werden dann in der angegebenen Reihenfolge in das Menü eingefügt. Die Elemente eines Menüs sind von einem der folgenden Typen:   

command      

ein Befehl, der mit der Option -command angegeben wird.        

checkbutton 

radiobutton entspricht dem gleichnamigen Widget. 

separator      

Trennlinie. Dieses Element kann nicht weiter konfiguriert werden. 

cascade     

 Untermenü 

Alle Elemente, abgesehen von separator-Einträgen, werden mit 
-label text   beschriftet.

Optionen für Menü-Elemente

Menü-Elemente verfügen im Wesentlichen über dieselben Optionen wie die analogen Widgets. Diese Optionen werden entweder beim Anlegen des Elements angegeben oder mit 

$menu entryconfigure index ?option? ?wert? ...   

definiert. Neben den üblichen Optionen 

-activebackground   farbe 

-activeforeground   farbe 

-background (-bg)  farbe 

-bitmap bitmap 

-command  befehl 

-font  font 

-foreground (-fg)  farbe 

-image image 

-state   normal|disabled 

-underline  index 

sind für alle Typen von Elementen noch folgende Optionen interessant:   

-accelerator abkürzung      
gibt rechts neben dem Label eine Tastaturabkürzung für den Aufruf an. Dies erzeugt nicht automatisch ein  entsprechendes Binding.        

-columnbreak  0|1      
Wenn die Option den Wert 1 hat, erscheint der Eintrag in einer neuen Spalte des Menüs. 

Beispiel

.mbar.cmd 

.mbar.cmd add command -label "Show" \

        -command { set Value "$Mbar_col $Mbar_cb1 $Mbar_cb2" }

.mbar.cmd add command -label "Config" \

        -command { .value config -fg $Mbar_fg -bg $Mbar_bg }

.mbar.cmd add command -label "Quit" -command exit -foreground red \

        -accelerator "(Strg-q)" -columnbreak true 

bind . <Control-q> { .mbar.cmd invoke "Quit" }

Check- und Radiobuttons

Checkbuttons und Radiobuttons werden in Menüs genauso verwendet wie die entsprechenden Widgets. Zusätzlich zu den oben genannten Optionen gibt es daher noch die Optionen   

-indicatoron  1 | 0 

-onvalue  wert 

-offvalue  wert 

-selectcolor  farbe 

-selectimage  image 

-value wert 

-variable  varname 

Beispiel

menu .mbar.rb -tearoff 1

.mbar.rb add radiobutton -label "Schwarz" -value "black" -variable Mbar_col

.mbar.rb add radiobutton -label "Blau"    -value "blue" -variable Mbar_col

.mbar.rb add radiobutton -label "Gruen"   -value "darkgreen" -variable Mbar_col 

menu .mbar.cb -tearoff 1

.mbar.cb add checkbutton -label "Unterstreichen" -variable Mbar_cb1

.mbar.cb add checkbutton -label "Schriftart" -onvalue "fett" \

                -offvalue "normal" -variable Mbar_cb2

Untermenüs

Mit dem Befehl 
$menu add cascade -label name -menu submenu?optionen?  wird ein Verweis auf ein Untermenü erzeugt. Das Untermenü muss ein Child-Widget des Menüs sein. 

Beispiel

menu .mbar.sub

.mbar.sub add cascade -label "Vordergrund" -menu .mbar.sub.fg

.mbar.sub add cascade -label "Hintergrund" -menu .mbar.sub.bg 

menu .mbar.sub.fg 

.mbar.sub.fg add radiobutton -label "Weiss" -value "white"  -variable Mbar_fg

.mbar.sub.fg add radiobutton -label "Gelb"  -value "yellow" -variable Mbar_fg

.mbar.sub.fg add radiobutton -label "Cyan"  -value "cyan"   -variable Mbar_fg 

menu .mbar.sub.bg

.mbar.sub.bg add radiobutton -label "Schwarz" -value "black" -variable Mbar_bg

.mbar.sub.bg add radiobutton -label "Blau"    -value "blue" -variable Mbar_bg

.mbar.sub.bg add radiobutton -label "Gruen"   -value "darkgreen" -variable Mbar_bg

Tearoff-Menüs

Wenn ein Menü einen Tearoff-Eintrag hat, kann es durch Anklicken dieses Eintrages zu einem separaten Fenster gemacht werden, das bis zum Schliessen auf dem Bildschirm bleibt. Die in diesem Zusammenhang relevanten Optionen sind   

-tearoff  1|0      

gibt an, ob das Menü einen Tearoff-Eintrag haben soll.        

-tearoffcommand befehl  
Der Befehl wird ausgeführt, wenn der Tearoff-Eintrag aktiviert      wird.        

-title titel      

Der Titel wird angezeigt, wenn das Menü in einem eigenen   Fenster steht. 

Beispiel

	menu .mbar.sub -tearoff 1 -title Farben

.mbar.sub add cascade -label "Vordergrund" -menu .mbar.sub.fg

.mbar.sub add cascade -label "Hintergrund" -menu .mbar.sub.bg


Menubars

Um ein Menü aufrufen zu können, muss es mit dem Fenster, zu dem es gehört, verknüpft werden. Ein Weg, dies zu erreichen, ist die Angabe eines Menubars, der die Menüs als Untermenüs enthält: 

menu name -type menubar 

Ein Menubar kann nur für ein Frame oder Toplevel-Fenster eingerichtet werden. 

Das Aussehen eines Menubars ist plattform-abhängig. Unter UNIX wird z.B. ein Help-Menü immer nach rechts platziert. 

Beispiel

menu .mbar -type menubar -relief ridge

.mbar add cascade -label "Help"        -menu .mbar.help

.mbar add command -label "Quit"        -command exit

.mbar add cascade -label "Radiobutton" -menu .mbar.rb

.mbar add cascade -label "Checkbutton" -menu .mbar.cb

.mbar add cascade -label "Cascade"     -menu .mbar.sub

pack .mbar -fill x -pady 6 

label .value -textvariable Value -bg white

pack .value -fill x -expand true -pady 6

Menubuttons

Eine andere Art Menüs aufzurufen besteht in der Anlage eines Menubuttons. Ein Menubutton entspricht einem gewöhnlichen Button, der jedoch keinen Befehl aufruft, sondern ein Menü darstellt. Dieses Menü muss ein Child-Widget des Menubuttons sein. 

Ein Menubutton wird durch 
menubutton name -menu menu ?optionen? erzeugt. 

Als Optionen können die Optionen des button-Widgets ( mit Ausnahme von -command ) sowie die folgenden spezifischen Optionen angegeben werden:   

-direction  below|above|left|right|flush  
gibt an, wo am Menubutton das Menü dargestellt wird. flush bedeutet, das es auf dem Button dargestellt wird.        

-indicatoron  0|1      

gibt an, ob ein Indikator für ein Menü neben den Text gesetzt werden soll.   

Beispiel

menubutton .mb -text Hauptmenue -menu .mb.menu -relief raised -indicator on   

menu .mb.menu

.mb.menu add cascade 
-label "Help"        

-menu .mb.menu.help

.mb.menu add command 
-label "Quit"        

-command exit -foreground red

.mb.menu add cascade
-label "Radiobutton" 
-menu .mb.menu.rb

.mb.menu add cascade 
-label "Checkbutton" 
-menu .mb.menu.cb

.mb.menu add cascade 
-label "Cascade"     

-menu .mb.menu.sub

Options-Menüs

Menüs werden häufig dazu verwendet, um aus einer Anzahl von Optionen auszuwählen. Daher gibt es für diesen Zweck eine eigene Tk-Prozedur: 

tk_optionMenu name variable option1 ?option2 ...? 

Sie erzeugt einen Menubutton name mit einem zugehörigen Radiobutton-Menü name.menu, aus dem eine der angegebenen Optionen ausgewählt werden kann. Diese Option wird auch in der Variablen variable gespeichert. Der Menubutton wird mit der jeweils ausgewählten Option beschriftet. 

Beispiel

tk_optionMenu .options Option option1 option2 option3

pack .options

.options.menu config -fg blue


